
global user "`c(username)'"
global dirdata "C:\Users\\$user\Dropbox\CCT BJP\data"
global dir1  "$dirdata\Final data"
global dir2  "$dirdata\Results\graphs"
global dir3  "$dirdata\Results\"
cd "$dir3"

use "$dir1\pooled_hh.dta", clear
keep if tau>=-4  
g age=age_m if hhmale==0
replace age=age_p if hhmale==1
g spanish=spanish_m if hhmale==0
replace spanish=spanish_p if hhmale==1

g Post=tau>=0
egen w_hoursw_head=rowtotal(w_hoursw_p w_hoursw_m)
g work_head=w_hoursw_head>0
replace work_head=. if work_m==. & work_p==. 
egen w_ylabor_head=rowtotal(w_ylabor_p w_ylabor_m)

local sample1 "(min_schooling>=1 & min_schooling<9)"
local nwork_lab "# of working adults"
local w_hoursw_t_lab "Hours/week"
local work_head_lab "Worked last week"
local w_hoursw_head_lab "Hours/week"
local work_m_lab "Worked last week"
local work_p_lab "Worked last week"
local w_hoursw_m_lab "Hours/week "
local w_hoursw_p_lab "Hours/week "
local parm Post
local replace replace

g taun5=tau==-5
global covs4 "age spanish schooling hhmale  urban np nchild5  hhmale age_sac  c.age_sac#i.year c.np#i.year i.departamento#i.year i.year"


local spec0  Post i.min_schooling  i.year  
local spec4  Post i.min_schooling  $covs4  

quietly {
foreach x in  work_head w_hoursw_head w_ylabor_head nwork w_hoursw_t w_ylabor_t {
sum `x' [w=factor]  if tau==-1 & `sample1'
local mean_dv=r(mean)
areg `x'  `spec0' [w=factor]  if `sample1', a(departamento) cluster(municipio)
outreg2 using "$dir3\Table2", excel `replace' ctitle(``x'_lab') keep(`parm') less(0) nocons  bdec(3) adds(Clusters, e(N_clust), Mean DV, `mean_dv') alpha(0.01, 0.05, 0.1) addtext(Demographic Controls, NO, Avg. age X year FE, NO, HH size X year FE, NO, Region X year FE, NO)   
local replace append
areg `x'  `spec4' [w=factor] if `sample1', a(departamento) cluster(municipio)
outreg2 using "$dir3\Table2", excel `replace' ctitle(``x'_lab') keep(`parm') less(0) nocons  bdec(3) adds(Clusters, e(N_clust),Mean DV, `mean_dv') alpha(0.01, 0.05, 0.1) addtext(Demographic Controls, YES, Avg. age X year FE,  YES, HH size X year FE, YES, Region X year FE, YES )   
}
}
